AC Lessons in electric circuits

27/8/18

Chapter 11 Power factor

1



$$Z = 60\,\Omega + j\,50{,}27\,\Omega = 78{,}28\,\Omega \angle 40°$$

$$I = \frac{U}{Z} = \frac{230V \angle 0°}{78{,}28\,\Omega \angle 40°} = 2{,}94A \angle -40°$$

$$P = I^2 \cdot R = (2{,}94A)^2 \cdot 60\,\Omega = 518{,}62\,W$$

$$Q = I^2 X_L = (2{,}94A)^2 \cdot 50{,}27\,\Omega = 434{,}51\,VAR$$

$$S = U \cdot I = 230V \cdot 2{,}94A = 676{,}2\,VA$$

$$\text{Power factor} = \cos 40° = 0{,}766 = \frac{P}{S}$$

27/8/18

2

Power factor correction with parallel capacitor

$$Q = \frac{U^2}{X_C} \Rightarrow X_C = \frac{U^2}{Q} = \frac{(230V)^2}{434,51\,VAR} = 121,75\,\Omega$$

$$X_C = \frac{1}{\omega C} \Rightarrow C = \frac{1}{121,75\Omega \cdot 2\pi\, 50Hz} = 2,61 \cdot 10^{-5}\,F$$

$$C = 26,1\,\mu F$$



$$Z = X_C \,||\, (X_L + R) = \frac{121,75\Omega \angle -90^\circ \cdot 78,28\Omega \angle 40^\circ}{121,75\Omega \angle -90^\circ + 78,28\Omega \angle 40^\circ}$$

$$Z = \frac{9530,59\,\Omega^2 \angle -50^\circ}{0 - j\,121,75\Omega + 60\Omega + j\,50,27\Omega} = \frac{9530,59\,\Omega^2 \angle -50^\circ}{60\Omega - j\,71,48\Omega}$$

$$Z = \frac{9530,59\,\Omega^2 \angle -50^\circ}{93,32\Omega \angle -50^\circ} = 102,13\Omega \angle 0^\circ$$

I = U / Z = 230 V / 102,13 Ω = 2,25 A in phase with U

$I_C = U / X_C$ = 230 V / 121,75 Ω angle – 90º = 1,89 A angle 90º

$I_{LR} = U / Z_{LR}$ = 230 V / 78,28 Ω angle 40º = 2,94 A angle -40º

Circuit without reactive power compensation I = 2,94 A



Circuit with reactive power compensation I = 2,25 A



NETLIST without PF compensation

```
***160 mH -- 60 ohm ***

*Lessons in electric circuits AC - Toni Kuphaldt
*11 Power factor

*without Power factor compensation

vin 1 0 AC 230 sin (0 1 50); AC voltage source necessary for frequency analisys


*inductor, the first 2 numbers are the nodes, the third the inductance in mH
l 1 2 160m

*resistor, the first 2 numbers are the nodes, the third the resistance value in ohm
r 2 0 60

.control

*transient analisys
*manual 15.3.9 - tran tstep tstop <tstart <tmax>> <uic>
* tstep time increment, 1 ms
* tstop final time 100 ms
* tstart start time 60 ms to observe stable behaviour after 60 ms (= 3 periods)
tran 1ms 100ms 60ms

plot v(1)
plot -vin#branch; i(t)
plot v(1, 2); vL(t)
plot v(2); vR(t)

plot v(1) -vin#branch v(1, 2)*80 v(2)


*frequency analisys linear one frequency from 50 Hz to 50 Hz
ac lin 1 50 50

* v(1)- Voltage source - vm voltage magnitude and (vp angle in radians / 2pi) * 360, converted to degrees
print v(1) vm(1) ((vp(1)/6.283)*360)

* v(1, 2)- Voltage over L1 - vm voltage magnitude and (vp angle in radians / 2pi) * 360, converted to degrees
print v(1, 2) vm(1, 2) ((vp(1, 2)/6.283)*360)

* Total current I – rectangular coordinates
print -vin#branch;total current

* v(2)- Voltage over R1 - vm voltage magnitude and (vp angle in radians / 2pi) * 360, converted to degrees
print v(2) vm(2) ((vp(2)/6.283)*360)

.endc
.end
```

NGSPICE simulation result NETLIST 1 without PF compensation

frequency analisys 50 Hz

Source
v(1) = 230 V, angle 0
vm(1) = 230 V, angle 0

L1
v(1, 2) = 94.85 V + j 113 V (rectangular)
vm(1, 2) = 147.70 angle 50º (polar)

Current I
-vin#branch = 2,25 A - j 1,89 A (rectangular)
2,94 A angle -40º (polar)

R1
v(2) = 135,15 V – j 113,22 V (rectangular)
vm(2) = 176.31 V angle -40º (polar)

Circuit without reactive power compensation I = 2,94 A



NETLIST with PF compensation

```
***160 mH -- 60 ohm // 26,1 microF  ***

*Lessons in electric circuits AC - Toni Kuphaldt
*11 Power factor

*Power factor compensation

*voltage sources
vin 1 0 AC 230 sin (0 1 50);    AC voltage source necessary for frequency analisys
vLR 1 2 ac 0;                    dummy source for C current
vC 1 4 ac 0;                     dummy source for C current

*inductor, the first 2 numbers are the nodes, the third the inductance in mH
l 2 3 160m

*resistor, the first 2 numbers are the nodes, the third the resistance value in ohm
r 3 0 60

*capacitor, the first 2 numbers are the nodes, the third the capacitance value in microF
c 4 0 26.1u
****************************************************************************************************
.control

*manual 15.3.9 - tran tstep tstop <tstart <tmax>> <uic>
* tstep time increment, 1 ms
* tstop final time 100 ms
* tstart start time 60 ms to observe stable behaviour after 60 ms (= 3 periods)
tran 1ms 100ms 60ms

plot v(1) (-vin#branch)*80 v(2, 3) v(3) (vLR#branch)*70 (vC#branch)*60


*frequency analisys linear one frequency from 50 Hz to 50 Hz
ac lin 1 50 50

* v(1)- Voltage source - vm voltage magnitude and (vp angle in radians / 2pi) * 360, converted to degrees
print v(1) vm(1) ((vp(1)/6.283)*360)

* Total current I – rectangular coordinates
print -vin#branch; total current

* v(2, 3)- Voltage over L1 - vm voltage magnitude and (vp angle in radians / 2pi) * 360, converted to degrees
print v(2, 3) vm(2, 3) ((vp(2, 3)/6.283)*360)

* v(3)- Voltage over R1 - vm voltage magnitude and (vp angle in radians / 2pi) * 360, converted to degrees
print v(3) vm(3) ((vp(3)/6.283)*360)

* Current  iLR – rectangular coordinates
print vLR#branch

* Current  iC – rectangular coordinates
print vC#branch

.endc
.end
```

NGSPICE simulation result NETLIST 1 with PF compensation

frequency analisys 50 Hz

Source
v(1) = 230 V, angle 0
vm(1) = 230 V, angle 0

Current I
-vin#branch = 2,25 A - j -0,0011 A (rectangular)
2,25 A angle -0,03º (polar)

L1
v(2, 3) = 94.85 V + j 113.22 V (rectangular)
vm(2, 3) = 147.70 angle 50º (polar)

R1
v(2) = 135,15 V – j 113,22 V (rectangular)
vm(2) = 176.31 V angle -40º (polar)

Current LR
-vlr#branch = 2.25 A - j 1.89 A

Current C
-vc#branch = j 1.89 A

Circuit with reactive power compensation I = 2,25 A